tested 190817 using *SpinDynamica* 3.0.1 under *Mathematica* 11.0

initialization

```
Needs["SpinDynamica`"]
```

```
SetSpinSystem[2]
```

SetSpinSystem: the spin system has been set to $\{\{1, \frac{1}{2}\}, \{2, \frac{1}{2}\}\}$

Hamiltonian

- **CS Hamiltonians, including isotropic shifts and CSA**

```
H[j : Except[_List], opts___Rule] :=
  PeriodicFunction[t, 2 π / ωr,
   Evaluate[
    ((ωiso[j] + ωaniso[j] × {-η[j] / Sqrt[6], 0, 1, 0, -η[j] / Sqrt[6]}.
          WignerD[2, {0}] [{ΩPM[j], ΩMR, {αRL0 - ωr t, βRL, 0}}]) opI[j, "z"]
     ) /. {opts}
   ]
  ];
```

- **coupling Hamiltonian, including J-coupling and DD-coupling**

```
H[{j_, k_}, opts___Rule] :=
  PeriodicFunction[t, 2 π / ωr,
   Evaluate[
    (2 π J[{j, k}] opI[j].opI[k] + Sqrt[6] b[{j, k}] ×
        WignerD[2, {0, 0}] [{ΩPM[{j, k}], ΩMR, {αRL0 - ωr t, βRL, 0}}] × opT[{j, k}, {2, 0}]
     ) /. {opts}
   ]
  ];
```

- **total Hamiltonian**

```
Htot[opts___Rule] :=
 PeriodicFunction[t, Evaluate[(2 π / ωr) /. {opts}],
  Evaluate@Collect[(H[1][t] + H[2][t] + H[{1, 2}][t]) /. {opts},
    opI[___] | HoldPattern@Dot[opI[___] ...]]]
```

DefaultParameters

```
DefaultParameters = Sequence[
    ωiso[1] → 2 π (-5 × 10^3),
    ωaniso[1] → 2 π 2 × 10^3, η[1] → 0.5, ΩPM[1] → {0, 0, 0},
    ωiso[2] → 2 π 5 × 10^3,
    ωaniso[2] → 2 π 10 × 10^3, η[2] → 0.1, ΩPM[2] → {0, π / 2, 0},
    J[{1, 2}] → 60,
    b[{1, 2}] → 2 π × (-2) 10^3, ΩPM[{1, 2}] → {0, π / 2, 0},
    ΩMR → {0, π / 4, 0},
    αRL0 → 0, ωr → 2 π × 10 × 10^3, βRL → N@ArcTan@Sqrt[2]
  ];
```

```
Htot[DefaultParameters]
```

$$\text{PeriodicFunction}\Big[t,\ \frac{1}{10\,000},$$
$$2000\left(-4.91955\times10^{-17}+0.353553\,e^{-20\,000\,i\,\pi\,t}+0.353553\,e^{20\,000\,i\,\pi\,t}+0.125\,e^{-40\,000\,i\,\pi\,t}+0.125\,e^{40\,000\,i\,\pi\,t}\right)$$
$$\pi\left(I_1^{-}\bullet I_2^{+}+I_1^{+}\bullet I_2^{-}-4\left(I_{1z}\bullet I_{2z}\right)\right)+120\,\pi\left(I_{1x}\bullet I_{2x}+I_{1y}\bullet I_{2y}+I_{1z}\bullet I_{2z}\right)+$$
$$\Big(-10\,000\,\pi+4000\Big(-4.91955\times10^{-17}-0.353553\,e^{-20\,000\,i\,\pi\,t}-0.353553\,e^{20\,000\,i\,\pi\,t}+0.125\,e^{-40\,000\,i\,\pi\,t}+$$
$$0.125\,e^{40\,000\,i\,\pi\,t}-0.204124\left(0.102062+0.394338\,e^{-20\,000\,i\,\pi\,t}-0.105662\,e^{20\,000\,i\,\pi\,t}+\right.$$
$$\left.0.19045\,e^{-40\,000\,i\,\pi\,t}+0.0136737\,e^{40\,000\,i\,\pi\,t}\right)-0.204124\left(0.102062-0.105662\,e^{-20\,000\,i\,\pi\,t}+\right.$$
$$\left.0.394338\,e^{20\,000\,i\,\pi\,t}+0.0136737\,e^{-40\,000\,i\,\pi\,t}+0.19045\,e^{40\,000\,i\,\pi\,t}\right)\Big)\,\pi\Big)\,I_{1z}+$$
$$\Big(10\,000\,\pi+20\,000\Big(-4.91955\times10^{-17}+0.353553\,e^{-20\,000\,i\,\pi\,t}+0.353553\,e^{20\,000\,i\,\pi\,t}+0.125\,e^{-40\,000\,i\,\pi\,t}+$$
$$0.125\,e^{40\,000\,i\,\pi\,t}-0.0408248\left(0.102062-0.394338\,e^{-20\,000\,i\,\pi\,t}+0.105662\,e^{20\,000\,i\,\pi\,t}+\right.$$
$$\left.0.19045\,e^{-40\,000\,i\,\pi\,t}+0.0136737\,e^{40\,000\,i\,\pi\,t}\right)-0.0408248\left(0.102062+0.105662\,e^{-20\,000\,i\,\pi\,t}-\right.$$
$$\left.0.394338\,e^{20\,000\,i\,\pi\,t}+0.0136737\,e^{-40\,000\,i\,\pi\,t}+0.19045\,e^{40\,000\,i\,\pi\,t}\right)\Big)\,\pi\Big)\,I_{2z}\Big]$$

MAS spectrum

```
T = 80 × 10^-3; δt = 50 × 10^-6;
```

- **single orientation**

```
sig =
  Signal1D[{0, T, δt},
   BackgroundGenerator → Htot[DefaultParameters],
   ReportSignalCalculationMethod → True
  ];
```

Signal1D: Using SignalCalculationMethod → COMPUTE

Signal1D: the last sampling point has been dropped in order to get an even number of points.

Signal1D: Using LineBroadening → 2π × 18.3234 rad s⁻¹.

```
ListPlot[Re@FT@sig, Frame → True, Joined → True, PlotRange → All, Axes → None]
```



- **powder average**

```
sig =
  Signal1D[{0, T, δt},
   BackgroundGenerator → Htot[ΩMR → orientation, DefaultParameters],
   EnsembleAverage → {orientation, OrientationsAndWeights["ZCW144"]},
   CarouselAverage → True,
   ReportSignalCalculationMethod → True
  ];
```

Signal1D: Using SignalCalculationMethod → COMPUTE

Signal1D: the last sampling point has been dropped in order to get an even number of points.

Signal1D: Using LineBroadening → $2\pi \times 18.3234$ rad s^{-1}.

```
ListPlot[Re@FT@sig, Frame → True, Joined → True, PlotRange → All, Axes → None]
```



exchange of z-magnetization, with and without zero-quantum relaxation

- **single orientation, no relaxation**

```
T = 5 × 10^-3;
```

```
ΔIz = opI[1, "z"] - opI[2, "z"];
```

```
Δztraj =
 Trajectory[
  ΔIz -> ΔIz,
  {None, T},
  BackgroundGenerator → Htot[DefaultParameters],
  MaxSteps → Automatic
 ]
```

TrajectoryFunction[{{0, 5.×10^-3}} , <>]

```
Plot[Re@Δztraj[t], {t, 0, T}, Frame → True, PlotRange → All, Axes → None]
```



- **single orientation, with relaxation**

relaxation superoperator for uncorrelated fluctuating random fields, in slow-motion limit. Parameterize using the ZQ relaxation time TZQ

```
Γ[pars___Rule] := Superoperator[(-(TZQ^-1/2) ×
    (DoubleCommutationSuperoperator[opT[1, {1, 0}], opT[1, {1, 0}]] +
      DoubleCommutationSuperoperator[opT[2, {1, 0}], opT[2, {1, 0}]])) /. {pars}]
```

check the ZQ relaxation rate constant by evaluating the appropriate superoperator matrix element

```
-LiouvilleBracket[
  NormalizeOperator[opI[1, "+"].opI[2, "-"]],
  Γ[],
  NormalizeOperator[opI[1, "+"].opI[2, "-"]]
 ]
```

$$\frac{1}{\mathrm{TZQ}}$$

simulate the exchange of z-magnetization for several ZQ relaxation time constants.

```
ΔIz = opI[1, "z"] - opI[2, "z"];
```

- **TZQ → 500 10^-6**

```
T = 5 × 10^-3;
Δztraj =
 Trajectory[
  ΔIz → ΔIz,
  {None, T},
  BackgroundGenerator → CombineGenerators[Htot[DefaultParameters], Γ[TZQ → 500 × 10^-6]],
  MaxSteps → Automatic
 ]
```

```
TrajectoryFunction[ {{0, 5.×10^-3}} , <>]
```

```
Plot[Re@Δztraj[t], {t, 0, T}, Frame → True, PlotRange → All, Axes → None]
```



- **TZQ → 200 10^-6**

```
Δztraj =
 Trajectory[
  ΔIz → ΔIz,
  {None, T},
  BackgroundGenerator → CombineGenerators[Htot[DefaultParameters], Γ[TZQ → 200 × 10^-6]],

 ]
```

```
TrajectoryFunction[ {{0, 5.×10^-3}} , <>]
```

```
Plot[Evaluate[Re@Δztraj[t]], {t, 0, T}, Frame → True, PlotRange → All, Axes → None]
```



- **TZQ → 20 10^-6**

```
Δztraj =
 Trajectory[
  ΔIz → ΔIz,
  {None, T},
  BackgroundGenerator → CombineGenerators[Htot[DefaultParameters], Γ[TZQ → 20 × 10^-6]],

 ]
```

```
TrajectoryFunction[ {{0, 5.×10^-3}} , <>]
```

```
Plot[Evaluate[Re@Δztraj[t]], {t, 0, T}, Frame → True, PlotRange → {0, 1}, Axes → None]
```



- **TZQ → 2 10^-6**

```
Δztraj =
 Trajectory[
  ΔIz → ΔIz,
  {None, T},
  BackgroundGenerator → CombineGenerators[Htot[DefaultParameters], Γ[TZQ → 2 × 10^-6]],

 ]
```

```
TrajectoryFunction[ {{0, 5.×10^-3}} , <>]
```

```
Plot[Evaluate[Re@Δztraj[t]], {t, 0, T}, Frame → True, PlotRange → {0, 1}, Axes → None]
```



note that the magnetization exchange is quenched if the T2ZQ is very short

- **powder average, with relaxation**

relaxation superoperator for uncorrelated fluctuating random fields, in slow-motion limit. Parameterize using the ZQ relaxation time TZQ

```
T = 5 × 10^-3;
```

```
Γ[pars___Rule] := Superoperator[(-(TZQ^-1/2) ×
     (DoubleCommutationSuperoperator[opT[1, {1, 0}], opT[1, {1, 0}]] +
       DoubleCommutationSuperoperator[opT[2, {1, 0}], opT[2, {1, 0}]])) /. {pars}]
```

```
ΔIz = opI[1, "z"] - opI[2, "z"];
```

```
τr = (2 π / ωr) /. {DefaultParameters};
nr = Ceiling[T / τr]
```

```
50
```

- **T2ZQ = 1ms**

```
Δztraj =
 Trajectory[
  ΔIz → ΔIz,
  {None, T},
  BackgroundGenerator →
   CombineGenerators[
    Htot[ΩMR → orientation, DefaultParameters],
    Γ[TZQ → 10^-3]],
  EnsembleAverage → {orientation, OrientationsAndWeights["ZCW50"]}
 ]
```

```
TrajectoryFunction[ {{0, 5.×10^-3}} , <>]
```

```
Plot[Evaluate[Re@Δztraj[t]], {t, 0, T}, Frame → True, PlotRange → All, Axes → None]
```



- **T2ZQ = 500 μs**

```
Δztraj =
 Trajectory[
  ΔIz → ΔIz,
  {None, T},
  BackgroundGenerator →
   CombineGenerators[
    Htot[ΩMR → orientation, DefaultParameters],
    Γ[TZQ → 500 × 10^-6]],
  ,
  EnsembleAverage → {orientation, OrientationsAndWeights["ZCW50"]}
 ]
```

```
TrajectoryFunction[ {{0, 5.×10^-3}} , <>]
```

```
Plot[Evaluate[Re@Δztraj[t]], {t, 0, T}, Frame → True, PlotRange → All, Axes → None]
```



- **T2ZQ = 200 μs**

```
Δztraj =
 Trajectory[
  ΔIz → ΔIz,
  {None, T},
  BackgroundGenerator →
   CombineGenerators[
    Htot[ΩMR → orientation, DefaultParameters],
    Γ[TZQ → 200 × 10^-6]],
  ,
  EnsembleAverage → {orientation, OrientationsAndWeights["ZCW50"]}
 ]
```

```
TrajectoryFunction[ {{0, 5.×10^-3}} , <>]
```

```
Plot[Evaluate[Re@Δztraj[t]], {t, 0, T}, Frame → True, PlotRange → {0, 1}, Axes → None]
```



- **T2ZQ = 20 μs**

```
Δztraj =
 Trajectory[
  ΔIz → ΔIz,
  {None, T},
  BackgroundGenerator →
   CombineGenerators[
    Htot[ΩMR → orientation, DefaultParameters],
    Γ[TZQ → 20 × 10^-6]],
  ,
  EnsembleAverage → {orientation, OrientationsAndWeights["ZCW50"]}
 ]
```

```
TrajectoryFunction[ {{0, 5.×10^-3}} , <>]
```

```
Plot[Evaluate[Re@Δztraj[t]], {t, 0, T}, Frame → True, PlotRange → {0, 1}, Axes → None]
```